Chapter 28.
Modelling Volatility by Elasticity

The CEV Family Explained and Explored

28.1 Introduction

Attempts to improve the treatment of volatility have been developed by many authors. This can be done in several contexts, and our focus here will be on improving the modelling of the volatility of equity options. Different methods can be used to treat volatility of FX and interest-rate options. Within the equity area, possible approaches include, but are not necessarily limited to:

(a) volatility as a deterministic function of the (stochastic) stock price ("level-dependent" models);
(b) stochastic volatility as a separate factor from the stock price;
(c) GARCH approaches.

This chapter addresses a subset of models of type (a), known as the "constant elasticity of variance" (CEV) models, which capture in a simple way the notion that the volatility increases as the asset price decreases. There are good reasons for considering this case:

(1) the model nicely captures the leverage effect associated with asset prices;
(2) there is research suggesting that the pricing of warrants in particular is improved by the use of this model;
(3) it is analytically tractable;
(4) it predicts skews and, with some additional adjustments for spreads, smiles;
(5) one member of the family is equivalent to a non-normal and non-log-normal interest rate model (CIR), allowing us to explore aspects of the interest-rate mathematics simultaneously with the equity mathematics.

This last point explains in part why this last chapter is placed directly after the discussion of elementary interest-rate models - the CEV theory applied to equity derivatives is closely related to the CIR model. Furthermore, placing the discussion here allows us to end this book at a point closely related to where we began, with a discussion of some of the effects that can be seen in implied volatility, in an application where it does make sense!

There are quite a few papers now published on elasticity models, some of them with minor glitches in them - part of the purpose of this chapter is to straighten out some of the technology and to give a comprehensive presentation. Also, some of the formulae employed for Call options or warrants turn out to be rather complicated. The general CEV Call option formula involves an infinite series of Gamma-functions, which is somewhat awkward to evaluate, even in *Mathematica*, let alone in C/C++ directly. There is an approximation for the cases of the SRCEV model (where the volatility varies as the inverse square root of the asset price - it is this model that is equivalent to the CIR interest-rate model without the mean-reversion) known as the Cox-Beckers approximation. Unfortunately there are at least three different and inconsistent representations of this formula in the literature.

One of the purposes of this chapter is to derive new simplified forms of the exact pricing formula. In the particular case of the SRCEV model, we give a simple formula involving an integration from 0 to π of the normal distribution, which is readily implementable in *Mathematica* or C/C++. While it may be possible to derive similar formulae for other elasticities, I have not done so, but the values of options can always be done directly in terms of integrals of Bessel functions. This can be coded up in C or C++ using standard routines, for example those given by Press *et al* (1992). Other methods for evaluation are given by Schroder (1989), which an interesting paper that among other things gives the clearest exposition, of the approximation of the non-central chi-square by normal distributions, that I have seen.

The running example considered in detail is that of a Call option - this is of particular interest in the development of improved warrant-pricing models.

The analysis presented suggests that at least part of the volatility structure observed in market prices may be captured by a simple level-dependent model, and one interesting issue relates to how to bolt on further stochasticity or GARCH behaviour to this type of approach. This approach, although simplistic, nicely captures the "leverage effect", so it should be considered as at least a part of the overall picture. The role of two-factor or GARCH models should perhaps be more properly defined as explaining those effects not captured by the type of model presented here.

28.2 CIR Interest Rates and Asset Elasticity Models

The type of stochastic process under consideration is probably more familiar in the CIR interest-rate model, as discussed in the previous chapter. The CIR world is governed by an interest-rate diffusion of the form

$$d r = a(b - r)\, d t + \sigma \sqrt{r}\, d z$$

This describes a mean-reverting process. It has nice analytical properties, such as the closed-form solution for bond options we have briefly outlined in the previous chapter. On the other hand, if we set $b = 0$, and $r \to S,\ a \to (q - r)$, we get a process of the form

$$d S = (r - q)\, S\, d t + \sigma \sqrt{S}\, d z$$

We can write (2) in the form

$$\frac{dS}{S} = (r - q)\,d\,t + \frac{\sigma}{\sqrt{S}}\,d\,z$$

which is an equity-like process with a volatility varying as the inverse square root of the stock price. This is commonly known as an SRCEV model - "Square root constant elasticity of variance". There are more general CEV models, where the process takes the form

$$\frac{dS}{S} = (r - q)\,d\,t + \frac{\sigma}{S^{1-\alpha}}\,d\,z$$

where α is between 0 (normal) and 1 (log-normal). So we see that a special form of the CIR interest-rate model is equivalent to a special case of stochastic volatility model for the equity world. The only difference is in the details and instruments valued - options on bonds (payoffs exponential functions of r) cf. options on S (payoffs typically piecewise linear).

A brief review of the asset versions of these models was done recently by their principal inventor, J. Cox, (1996) in an article entitled: "The constant elasticity of variance option pricing model". It is well worth a detailed read, but note that the formula for the probability distribution is missing a term. Cox goes on to develop the infinite series of Gamma-functions as the price of a Call - it is this that we wish to try to simplify to make it more straightforward as a model. We shall also look at the associated Black-Scholes PDE in some detail, and explore its solution for a case slightly more general than that considered by Cox, where we have a time-dependent interest-rate and yield structure. Our approach will be based on the partial differential equation method, and parallels that given for the ordinary Black-Scholes equation, and it is here that the detailed development begins. Readers may wish to explore the development of an equivalent martingale formulation, and of the corresponding interest-rate analogues with mean reversion.

28.3 Standardized Form of the CEV Black-Scholes Equation

The Black-Scholes equation with time-dependent interest rates and dividend yield, and volatility $\sigma(S)$, is

$$\frac{\partial V}{\partial t} - r(t)\,V + \frac{1}{2}\,\sigma^2(S)\,S^2\,\frac{\partial^2 V}{\partial S^2} + (r(t) - q(t))\,S\,\frac{\partial V}{\partial S} = 0$$

First we make the standard time-reversing transformation $t^* = T - t$, with the option expiring at time T. This gives

$$\frac{\sigma^2(S)}{2}\,S^2\,\frac{\partial^2 V}{\partial S^2} + (r(T - t^*) - q(T - t^*))\,S\,\frac{\partial V}{\partial S} = \frac{\partial V}{\partial t^*} + r(T - t^*)\,V$$

Next we make the standard discounting transformation (Section 4.6) $V = e^{-B(t^*)}\,V_1$ with B arranged so that

$$\frac{\partial B}{\partial t^*} = r(T - t^*)$$

which reduces the equation to

$$\frac{\sigma^2(S)}{2} S^2 \frac{\partial^2 V_1}{\partial S^2} + (r(T - t^*) - q(T - t^*))\, S \frac{\partial V_1}{\partial S} = \frac{\partial V_1}{\partial t^*}$$

Note that the appropriate solution to the differential equation for B is indeed the discount factor:

$$e^{-B(t^*)} = e^{-\int_t^T r(t_1)\, dt_1}$$

The next task is to eliminate the drift term, and this again makes use of an approach similar to that discussed in Chapter 4 for CBs. We introduce new variables

$$S = S' \, e^{F(t')}$$
$$t^* = t'$$

and change variables in the PDE (just use the chain rule carefully if you want to check this), obtaining

$$\frac{\sigma^2(S)}{2} S'^2 \frac{\partial^2 V_1}{\partial S'^2} + \left(\frac{\partial F}{\partial t'} + r(T - t') - q(T - t')\right) S' \frac{\partial V_1}{\partial S'} = \frac{\partial V_1}{\partial t'}$$

Note that the volatility remains expressed in terms of the original underlying. So we can eliminate the drift term by choosing F to satisfy the ordinary differential equation

$$\frac{\partial F}{\partial t'} + r(T - t') - q(T - t') = 0$$

We can write the solution formally in terms of an integral of the spot rate and yield. We shall write the answer in terms of a function Q that is the reciprocal of $e^{F(t')}$, i.e.,

$$Q(t') = e^{+\int_0^{t'} (r(T-s)-q(T-s))\, ds}$$

Then

$$S = \frac{S'}{Q(t')}$$

The PDE is now

$$\frac{\sigma^2(S'/Q)}{2} S'^2 \frac{\partial^2 V_1}{\partial S'^2} = \frac{\partial V_1}{\partial t'}$$

Note that at maturity $S = S'$, whereas at initiation

$$S' = S\, e^{+\int_0^T (r(s)-q(s))\,ds}$$

It is difficult to proceed further without an assumption about the form of the volatility. It is here that we make the elasticity assumption:

$$\sigma(S) = \sigma_0\, S^{\alpha-1}$$

This allows us to collect the independent variables on both sides of the equation, obtaining

$$Q(t')^{2(\alpha-1)}\, \frac{\partial V_1}{\partial t'} = \frac{1}{2}\, \sigma_0^2\, S'^{2\alpha}\, \frac{\partial^2 V_1}{\partial S'^2}$$

We re-parametrize the elasticity variable by writing $2\alpha = 2 - 1/\nu$, so that the PDE is now

$$\frac{1}{Q(t')^{1/\nu}}\, \frac{\partial V_1}{\partial t'} = \frac{1}{2}\, \sigma_0^2\, S'^{2-1/\nu}\, \frac{\partial^2 V_1}{\partial S'^2}$$

We are almost at a recognizable PDE! It is just a matter of making one further transformation on each of the time and price variables. To simplify the right side, we introduce

$$X = S'^{1/\nu}$$

This gets us to

$$\frac{2\nu^2}{\sigma_0^2\, Q^{\frac{1}{\nu}}(t')}\, \frac{\partial V_1}{\partial t'} = X\, \frac{\partial^2 V_1}{\partial X^2} + (1-\nu)\, \frac{\partial V_1}{\partial X}$$

The left side is now simplified by picking a new time coordinate τ with the property that

$$d\tau = \frac{1}{2\nu^2}\, \sigma_0^2\, Q^{\frac{1}{\nu}}(t')\, dt'$$

This differential equation can be solved formally as

$$\tau = \frac{\sigma_0^2}{2\nu^2} \int_0^{t'} e^{\frac{1}{\nu}\int_0^s (r(T-s')-q(T-s'))\,ds'}\, ds$$

Note that this new time is zero at expiry. If r and q are constant, this simplifies to

$$\tau = \frac{\sigma_0^2}{2\nu(r-q)} \left(e^{\frac{(r-q)(T-t)}{\nu}} - 1\right)$$

which is the reciprocal of the parameter called c or k by Cox and other authors. Whether or not the r, q parameters are constant, we arrive at the standardized CEV equation in the form

$$\frac{\partial V_1}{\partial \tau} = X \frac{\partial^2 V_1}{\partial X^2} + (1-\nu) \frac{\partial V_1}{\partial X}$$

The elasticity parameter ν appears now in the first order term - this vanishes in this view of the model when $\nu = 1$, which is the special square root CEV model. This PDE is akin to that for purely radial diffusion in $-\nu$ dimensions! Students of probability may recognize it as that corresponding to a Bessel-squared process. We note that a futher transformation can reduce this to the ordinary diffusion equation with a non-zero drift - a fact that is of considerable importance, and that also applies to all of the interest-rate models, when source terms are also admitted.

In principle we could proceed to analyse this equation in much the same way as the ordinary one-dimensional diffusion equation. As the details of this are hard - rather beyond the scope of this text - we shall just quote the Green's function analogous to that given in Chapter 4. The Green's function corresponding to a delta-function source located at Y at time $\tau = 0$ is:

$$G(X, Y, \tau) = \frac{1}{\tau} e^{-\frac{(X+Y)}{\tau}} \left(\frac{X}{Y}\right)^{\nu/2} I_\nu\left(\frac{2\sqrt{X Y}}{\tau}\right)$$

If you wish to satisfy yourself that this is correct, then use *Mathematica* to check that it solves the differential equation, and use the fact that for large z,

$$I_\nu(z) \sim \frac{e^z}{\sqrt{2\pi z}}$$

to verify the behaviour for small τ. Normalization issues are discussed presently. Taking G as the Green's function allows us to write down the option valuation and interpret it in terms of risk-neutral valuation. Let the option payoff be $P(S_T)$. Because $S' = S$ at maturity, the value of X at maturity, denoted by X_T, corresponding to the stock price S_T is just

$$S_T = X_T^\nu$$

The non-discounted value at an earlier time τ is then

$$V'(X, \tau) = \int P(X_T^\nu)\, G(X, X_T, \tau)\, dX_T$$

Making a change of variable to S_T, we are led, finally, to

$$V'(X, \tau) = \int P(S_T)\, f(S_T)\, dS_T$$

where, if we set $X = x\,\tau$, $X_T = x_T\,\tau$,

$$f(S_T) = \frac{1}{\nu}\,\tau^{-\nu}\left(x\,x_T^{\frac{2}{\nu}-3}\right)^{\nu/2}\,e^{-x-x_T}\,I_\nu\left(2\sqrt{x\,x_T}\right)$$

The variable X is then related to the price at initiation by

$$X = S^{1/\nu}\,e^{\frac{\int_0^T (r(s)-q(s))\,ds}{\nu}}$$

The one catch in dealing with such models is that the expiry distribution does not integrate to unity. The density function for X_T is

$$U(X_T) = \frac{1}{\tau}\,e^{-\frac{(X+X_T)}{\tau}}\left(\frac{X}{X_T}\right)^{\nu/2}\,I_\nu\left(\frac{2\sqrt{X\,X_T}}{\tau}\right)$$

This function has the property that

$$\int_0^\infty U(X_T)\,dX_T = 1 - \frac{\Gamma(\nu,\,\frac{X}{\tau})}{\Gamma(\nu)}$$

So the probability that $X_T = 0$ is just

$$\frac{\Gamma(\nu,\,\frac{X}{\tau})}{\Gamma(\nu)}$$

This has to be taken into account when valuing options whose payoff is non-zero at the origin - a Put being the most obvious case. This distribution is exceptional in having both a discrete portion and a continuous portion. There are many other interesting probabilistic aspects to CEV models, which we shall not consider here. A final reminder - the special cases are:

(a) $\nu = 1/2$ - the absolute Gaussian model;
(b) $\nu = 1$ - the SRCEV model;
(c) $\nu = \infty$ - the log-normal model.

28.4 CEV Call Option Formulae

By risk-neutral valuation, the value of a European Call with strike K is just

$$e^{-r\,t}\int_K^\infty (S_T - K)\,f(S_T)\,dS_T$$

There are no known "nice" formulae for this. Cox (1996) gives the result as an infinite series of Gamma-functions. Cox's model is the standard exact solution representation, and can be coded up in *Mathematica*, but requires careful treatment with respect to both evaluation of the Gamma-functions and also summation of the infinite series. A more down-to-earth representation is the direct integration approach involving the Bessel functions (use the routines in Numerical Recipes if working in C) and the evaluation of (36). The integral can be simplified a little in terms of a standard integral. If we define:

$$\text{CEVfunc}(x, a, \nu, n) = \int_a^\infty e^{-\frac{z^2}{4x}} z^n I_\nu(z)\, dz$$

then the value of a Call, with strike K, $c = \tau^{-1}$, $x = X/\tau$, is given by

$$e^{-rt-x} \left(\frac{\text{CEVfunc}\left(x, 2\sqrt{c\,x\,K^{1/\nu}}, \nu, \nu+1\right)}{x\,c^\nu\,2^{\nu+1}} - K\,(2\,x)^{\nu-1}\,\text{CEVfunc}\left(x, 2\sqrt{c\,x\,K^{1/\nu}}, \nu, 1-\nu\right) \right)$$

It is left to the reader to explore an interesting set of special cases when the object CEVfunc can be evaluated in closed-form. This occurs when ν is half an odd integer (cf Schroder, 1989), and provides a useful basis for approximating the answer by interpolating bewteen the values obtained for half-odd-integer ν. Our interest will focus on the case of general ν, and the particular case $\nu = 1$.

■ *Mathematica* Implementation

```
CEVfunc[x_, a_, ν_, n_, upper_] :=
  NIntegrate[Exp[-(z^2/(4*x))]*z^n*BesselI[ν, z], {z, a, upper}]
```

■ Call Formula

What we do is to directly integrate to a multiple of the strike, set by the variable "scale". Note that the σ parameter is set by assuming the volatility is that for the current spot price.

```
CEVCall[S_, K_, r_, q_, vol_, t_, ν_, scale_] :=
Module[{σ = vol * S^(1 / (2 ν)), x, c, a},
    c = (2 * ν * (r - q)) / (σ^2 * (Exp[((r - q) * t) / ν] - 1));
    x = c * S^(1 / ν) * Exp[((r - q) * t) / ν];
    a = 2 Sqrt[c x K^(1 / ν)];
        Exp[-r * t - x] *
   (CEVfunc[x, a, ν, ν + 1, scale * a] / (x * c^ν * 2^(ν + 1)) -
  K * ((2 * x)^(ν - 1)) * CEVfunc[x, a, ν, 1 - ν, scale * a])]
```

Here it is evaluated for a high value of $\nu = 10$, which should be close to the Black-Scholes value:

```
CEVCall[100, 100, 0.1, 0, 0.2, 1, 10, 10]
```

```
13.269711019
```

Is it close?

```
Needs["Derivatives`BlackScholes`"]
```

```
BlackScholesCall[100, 100, 0.2, 0.1, 0, 1]
```

```
13.2697
```

The answer is yes. As we lower the parameter the value increases for these at-the-money examples:

```
CEVCall[100, 100, 0.1, 0, 0.2, 1, 5, 10]
```

```
13.2698143336
```

Here is the SRCEV case:

```
CEVCall[100, 100, 0.1, 0, 0.2, 1, 1, 10]
```

```
13.2731
```

28.5 A Simplified Exact Model for SRCEV

There are special cases where the integral can be simplified significantly. In addition to the case of ν equal to half an odd integer, where closed-form answers exist, I believe a useful simplification can be obtained for integer values also, and the following explores the simplest case of $\nu = 1$, corresponding to the SRCEV model. The evaluation can be simplified dramatically by expressing the Bessel functions using an integral identity, and then doing the integration above. This still leaves an integral. However, the integrand now involves much simpler functions, and the remaining integral is typically over a simple finite interval such as $[0, \pi]$. The integration of periodic functions over such intervals is particularly simple, due to the special nature of the numerical error formula for periodic integrals. It can be evaluated to any desired degree of precision, but in practice only a few samples are needed. For example, in the case $\nu = 1$, we can write

$$I_1(z) = \frac{1}{\pi} \int_0^{\pi} \cos(x)\, e^{z \cos(x)}\, dx$$

Integral identities similar to this do exist for other values of ν. The difficult integral from the strike to infinity can be done in closed form in terms of the standard cumulative Normal distribution. The result is as follows - let

$$\alpha = 2\, c\, K;\ \beta = 2\, x;$$

$$f(\alpha, \beta, \theta) = (\beta \cos^2(\theta) - \alpha + 1)\left(1 - N\left(\sqrt{\alpha} - \cos(\theta)\sqrt{\beta}\right)\right)\sqrt{\frac{\beta}{2\pi}} + \frac{\sqrt{\beta}\, e^{-\frac{1}{2}\left(\sqrt{\alpha} - \cos(\theta)\sqrt{\beta}\right)^2}\left(\sqrt{\alpha} + \cos(\theta)\sqrt{\beta}\right)}{2\pi}$$

Then the value of a call option is just

$$\frac{1}{c}\, e^{-r t - \frac{\beta}{2}} \int_0^{\pi} \cos(\theta)\, e^{\frac{1}{2}\beta \cos^2(\theta)} f(\alpha, \beta, \theta)\, d\theta$$

Let's first build a *Mathematica* implementation:

```
interfunc[α_, β_, θ_] :=
 Sqrt[β / (2 Pi)] * (1 - α + β Cos[θ] ^ 2) (1 - Norm[Sqrt[α] - Cos[θ] * Sqrt[β]]) +
  (Sqrt[β] / (2 * Pi)) * (Sqrt[α] + Cos[θ] * Sqrt[β]) *
   Exp[-1 / 2 (Sqrt[α] - Cos[θ] * Sqrt[β]) ^ 2]

SRCEVCall[S_, K_, r_, q_, σ_, t_] :=
    Module[{sig = σ * Sqrt[S], α, β, c, int},
        c = 2 * (r - q) / (sig ^ 2 * (Exp[(r - q) * t] - 1));
        α = 2 * c * K;
        β = 2 * c * S * Exp[(r - q) * t];
        int =
   NIntegrate[Cos[θ] Exp[β Cos[θ] ^ 2 / 2] interfunc[α, β, θ], {θ, 0, Pi}];
        Exp[-r * t - β / 2] * int / c]

SRCEVCall[100, 100, 0.1, 0, 0.2, 1]

      13.2731
```

This evaluates rather more quickly, and rather than involving Bessel functions and an infinite integration, involves only the normal distribution and a finite integration. Because of the nature of periodic functions, the integration can be done very efficiently using the trapezoidal rule and relatively few points. We halve the integration done from $[0,\ 2\pi]$ and proceeding by just summing the values of the integrand sampled at $(0,\ 2\pi/n,\ 4\pi/n,\ \ldots k*2\pi/n,\ \ldots 2\pi(1-1/n))$:

```
NumericalSRCEVCall[S_, K_, r_, q_, σ_, t_, samples_] :=
    Module[{sig = σ * Sqrt[S], α, β, c, sum},
        c = 2 * (r - q) / (sig ^ 2 * (Exp[(r - q) * t] - 1));
        α = 2 * c * K;
        β = 2 * c * S * Exp[(r - q) * t];
        sum = (2 Pi) / samples Sum[Cos[θ] Exp[β Cos[θ] ^ 2 / 2] interfunc[α, β, θ],
     {θ, 0, 2 * Pi (1 - 1 / samples), 2 Pi / samples}];
        Exp[-r * t - β / 2] * sum / (2 * c)]
```

60 terms are enough to get agreement to six decimal places, and it evaluates very quickly.

```
NumericalSRCEVCall[100, 100, 0.1, 0, 0.2, 1, 60]
```

```
13.2731
```

This is therefore a trivial addition to any derivatives modelling system where the cumulative normal distribution is already coded up, and is exact within the framework of the model - any desired degree of precision may be obtained by taking more terms in the summation.

28.6 The Cox-Beckers Approximation to SRCEV

This just involves a nest of substitutions and what looks like a normal approximation to the non-central chi-square distribution whose pdf we have given. It is just a bunch of formulae, but implementation is complicated by the fact that the literature has, what appears to this author at least, three different and inconsistent representations of this formula. If you wish to make your own evaluation of these comments, four papers all containing a version of this approximation are Beckers (1980), Jarrow and Rudd (1983), Lauterbach and Schultz (1990), Hauser and Lauterbach (1996). In the following, y is essentially c times the underlying. The implementation in here is based on Jarrow and Rudd's (1983) book, which in the humble opinion of this author is the correct version, but this is based on the subsequent comparison with the exact solution. I cannot claim to have gone into the approximation of the required distributions by normal distributions in any detail, but the following appears also to be consistent with the remarks by Schroder (1989) who also gives more general results that apply for non-zero yield. In any case it is probably more efficient to use the exact new model given above, and the following is largely included for completeness. Here are the formulae:

```
y[S_, r_, σ_, t_] := 4*r*S/(σ^2 (1 - Exp[-r*t]));
z[K_, r_, σ_, t_] :=  4*r*K/(σ^2 (Exp[r*t] - 1));
p[v_, y_] := (v + 2 y) / (v + y) ^2;
h[v_, y_] := 1 - 2*(v + y)*(v + 3 y) / 3 / (v + 2 y) ^2;
q[v_, h_, p_, z_, y_] :=
  (1 + h*(h - 1)*p  - 1/2 h (h - 1) (2 - h) (1 - 3 h) p^2 - (z / (v + y)) ^h) /
   Sqrt[2*h^2 p (1 - (1 - h) (1 - 3 h) p)];
```

The Call formula is then

$$S\,N(q(4,\,h(4,\,Y),\,p(4,\,Y),\,Z,\,Y)) - X\,e^{-r\,t}\,N(q(0,\,h(0,\,Y),\,p(0,\,Y),\,Z,\,Y))$$

The *Mathematica* representation of this is

```
CBAppSRCEVCall[S_, X_, r_, σ_, t_] :=
 Module[{sigone, Y, Z, Pzero, Pfour, Hzero, Hfour, Qzero, Qfour},
        sigone = σ*Sqrt[S]; Y = y[S, r, sigone, t];
  Z = z[X, r, sigone, t]; Pzero = p[0, Y];     Pfour = p[4, Y];
        Hzero = h[0, Y]; Hfour = h[4, Y];
  Qzero = q[0, Hzero, Pzero, Z, Y];
        Qfour = q[4, Hfour, Pfour, Z, Y];
        S*Norm[Qfour] - X Exp[-r*t]*Norm[Qzero]]
```

■ Comparisons with Exact Solution

Let's look at the approximate solution vs the exact solution for ATM, ITM, OTM cases:

```
{CBAppSRCEVCall[100, 100, 0.1, 0.2, 1], SRCEVCall[100, 100, 0.1, 0, 0.2, 1]}

    {13.2763, 13.2731}

{CBAppSRCEVCall[110, 100, 0.1, 0.2, 1], SRCEVCall[110, 100, 0.1, 0, 0.2, 1]}

    {21.3727, 21.3699}

{CBAppSRCEVCall[90, 100, 0.1, 0.2, 1], SRCEVCall[90, 100, 0.1, 0, 0.2, 1]}

    {6.76843, 6.76697}
```

28.7 Cox's Gamma-Function Series

For completeness the exact solution given by Cox is presented. We can then compare this with the direct integration of the discounted risk-neutral expiry distribution. We need first to define some Gamma-functions.

```
?Gamma

    Gamma[z] is the Euler gamma function. Gamma[a,
      z] is the incomplete gamma function. Gamma[a, z0, z1] is the
      generalized incomplete gamma function Gamma[a, z0] - Gamma[a, z1].

g[n_, z_] := Exp[-z] z^(n-1)/Gamma[n]

G1[n_, w_] = Integrate[g[n,w], w]
```

$$-\frac{\text{Gamma}[n, w]}{\text{Gamma}[n]}$$

This *G* is the one defined by Cox - it is the complementary cumulative distribution function.

```
G[n_, w_] := Gamma[n, w]/Gamma[n]
```

Next we need a series as given by Cox:

```
CoxSeries[a_, b_, x_, y_, m_] := Sum[g[n+a,x]*G[n+b,y], {n, 1, m}]
```

```
PartialCoxSeries[a_, b_, x_, y_, kk_, m_] := Sum[g[n+a,x]*G[n+b,y], {n, kk, m}]
```

Here is what the series looks like for a few terms:

```
CoxSeries[a,b,x,y,2]
```

$$\frac{E^{-x}\, x^{a}\, \text{Gamma}[1+b,\, y]}{\text{Gamma}[1+a]\,\text{Gamma}[1+b]} + \frac{E^{-x}\, x^{1+a}\, \text{Gamma}[2+b,\, y]}{\text{Gamma}[2+a]\,\text{Gamma}[2+b]}$$

In the following formula the elasticity parameter corresponds to α, and the yield is zero.

```
SeriesCEVOption[p_, k_, vol_, r_, t_, elas_, trunc_] :=
Module[{lambda = 0.5/(1-elas), sig = vol*p^(1-elas), x, y, rtl},
rtl = Exp[r(t/lambda)];
x = 2 lambda r p^(1/lambda) rtl/((rtl - 1)*sig^2);
y = 2 lambda r k^(1/lambda)/((rtl - 1)*sig^2);
p*CoxSeries[0,lambda, x, y, trunc] - k*Exp[-r*t]*CoxSeries[lambda,0,x,y, trunc]]
```

```
SeriesCEVOption[S, K, σ, r, t, e, n]
```

$$-E^{-0.06\,t}\, K \sum_{n=1}^{n} g\Big[n + \frac{0.5}{1-e},\, \frac{0.06\, E^{0.12\,(1-e)\,t}\, S^{-2+2.\,(1-e)+2\,e}}{(1-e)\,(-1+E^{0.12\,(1-e)\,t})\,\sigma^2}\Big]\; G\Big[n+0,\, \frac{0.06\, K^{2.\,(1-e)}\, S^{-2+2\,e}}{(1-e)\,(-1+E^{0.12\,(1-e)\,t})\,\sigma^2}\Big] + S \sum_{n=1}^{n} g\Big[n+0,\, \frac{0.06\, E^{0.12\,(1-e)\,t}\, S^{-2+2.\,(1-e)+2\,e}}{(1-e)\,(-1+E^{0.12\,(1-e)\,t})\,\sigma^2}\Big]\; G\Big[n+\frac{0.5}{1-e},\, \frac{0.06\, K^{2.\,(1-e)}\, S^{-2+2\,e}}{(1-e)\,(-1+E^{0.12\,(1-e)\,t})\,\sigma^2}\Big]$$

With enough terms in the series, we can compute a stable result. Here we give the SRCEV values.

```
SeriesCEVOption[100, 100, 0.20, 0.1, 1, 0.5, 140]
```

```
13.2731
```

Just to check that we have enough terms:

```
SeriesCEVOption[100, 100, 0.20, 0.1, 1, 0.5, 150]
```

```
13.2731
```

Now we can look closer to Black-Scholes, by increasing the elasticity to be closer to unity. This requires very many more terms. The formula is singular when the elasticity is unity, so we set it to 0.95, to compare with the Bessel function form. Counting in thousands, 6000 terms are both necessary and sufficient for 6 SF:

```
SeriesCEVOption[100, 100, 0.20, 0.1, 1, 0.95, 5000]

    1.69942

SeriesCEVOption[100, 100, 0.20, 0.1, 1, 0.95, 6000]

    13.2697

SeriesCEVOption[100, 100, 0.20, 0.1, 1, 0.95, 7000]

    13.2697
```

28.8 Skews and Smiles by "Real-World" Option Pricing

The idea in this section is to explore what happens when the price computed by an allegedly more realistic model (here a CEV model) is fed backwards through Black-Scholes, to work out the implied volatility. The point to be made is that there is a simple mechanism for the emergence of a skew, based on an elasticity approach. Note that this is not due to the volatility going up as the price at initiation goes down, for we are fixing the volatility as a fixed constant at time zero. The skew arises as a result of subsequent evolution of the volatility as the asset price changes.

```
Needs["Derivatives`BlackScholes`"]

? BlackScholesCallImpVol

    BlackScholesCallImpVol[price, strike, riskfree, divyield, expiry,
      optionprice]  returns the implied volatility of a vanilla European Call.
```

So we fix the volatility, price with the SRCEV model, then infer the implied volatility from the Black-Scholes model. As before, the σ parameter is set simply according to the scale of variation of the underlying.

```
ForwardBackward[S_, K_, r_, q_, σ_, t_] :=
    BlackScholesCallImpVol[S, K, r, q, t, SRCEVCall[S, K, r, q, σ, t]]

{ForwardBackward[100, 90, 0.1, 0, 0.2, 1],
 ForwardBackward[100, 100, 0.1, 0, 0.2, 1],
 ForwardBackward[100, 110, 0.1, 0, 0.2, 1]}

    {0.20538, 0.200104, 0.195409}
```

■ A Skew or Half a Smile

This model does capture a volatility skew typical of major indices under certain circumstances - the volatility at time zero is always 20%, but the volatility implied from the BS model varies from nearly 22% with a strike at 70, to under 19% with a strike at 130. Note that the out-of-the-money end is at the right of this plot.

```
Plot[ForwardBackward[100, K, 0.05, 0, 0.2, 1], {K, 70, 130}];
```



28.9 The Effect of Spread

Real-world trading can also involve a bid-offer spread. Suppose we add 2.5% to the elasticity result and again compute the implied volatility:

```
ForwardBackward[S_, K_, r_, q_, σ_, t_, spread_] :=
    BlackScholesCallImpVol[S, K,
  r, q, t, (1 + spread / 100) * SRCEVCall[S, K, r, q, σ, t]]

Plot[ForwardBackward[100, K, 0.05, 0, 0.2, 1, 2.5], {K, 70, 130},
    PlotLabel -> "Skew for Elasticity\n+ Spread Model of Call\n"];
```



28.10 Puts and Smiles

The discussion of Puts is complicated by the fact that there is a non-zero probability of the stock price reaching zero and staying there, for finite values of ν. For example, in our test problem, it is very small:

```
ZeroProb[S_, r_, q_, vol_, t_, ν_] :=
Module[{σ = vol * S ^ (1 / (2 ν)), x, c},
    c = (2 * ν * (r - q)) / (σ^2 * (Exp[((r - q) * t) / ν] - 1));
    x = c * S ^ (1 / ν) * Exp[((r - q) * t) / ν];
Gamma[ν, x] / Gamma[ν]]
```

```
ZeroProb[100, 0.05, 0, 0.2, 1, 1]
```

5.4687×10^{-23}

As ν increases (i.e. we move closer to standard log-normal) it becomes smaller:

```
ZeroProb[100, 0.05, 0, 0.2, 1, 10]
```

$6.85331107725 \times 10^{-2150}$

It can be significant in the Gaussian case, for longer times and smaller drifts:

```
ZeroProb[100, 0.02, 0, 0.2, 5, 0.5]
```

0.0188362

What we shall do is model a Put which includes the contribution for the final asset price being zero, so that it only remains to consider the continuous part of the distribution, which proceeds as before. (You should check that this ensures Put-Call parity applies.) The model then becomes the pair of functions

```
CEVPutfunc[x_, a_, ν_, n_] :=
  NIntegrate[Exp[-(z^2/(4*x))]*z^n*BesselI[ν, z], {z, 0, a}]
```

```
CEVPut[S_, K_, r_, q_, vol_, t_, ν_] :=
Module[{σ = vol * S^(1 / (2 ν)), x, c, a},
    c = (2 * ν * (r - q)) / (σ^2 * (Exp[((r - q) * t) / ν] - 1));
    x = c * S^(1 / ν) * Exp[((r - q) * t) / ν];
    a = 2 Sqrt[c x K^(1 / ν)];
        Exp[-r * t - x] (K * ((2 * x)^(ν - 1)) * CEVPutfunc[x, a, ν, 1 - ν] -
      CEVPutfunc[x, a, ν, ν + 1] / (x * c^ν * 2^(ν + 1)) ) +
   K * Exp[-r * t] * ZeroProb[S, r, q, vol, t, ν]]
```

```
CEVPut[100, 100, 0.05, 0, 0.2, 1, 1]
```

```
5.57683
```

As before, as ν gets large we approach the BS values:

```
BlackScholesPut[100, 100, 0.2, 0.05, 0, 1]
```

```
5.57353
```

```
CEVPut[100, 100, 0.05, 0, 0.2, 1, 100]
```

```
5.5735264
```

We can do corresponding implied volatility computations, limiting attention, as before, to SRCEV:

```
PutForwardBackward[S_, K_, r_, q_, σ_, t_, spread_] :=
    BlackScholesPutImpVol[S, K, r, q,
  t, (1 + spread / 100) * CEVPut[S, K, r, q, σ, t, 1 / 2]]
```

With no spread, the skew obtained is as follows - note that the out-of-the-money case is at the left of this plot:

```
Plot[PutForwardBackward[100, K, 0.05, 0, 0.2, 1, 0], {K, 70, 130}];
```



This time folding in the spread effect turns the skew into the beginning of a smile.

```
Plot[PutForwardBackward[100, K, 0.05, 0, 0.2, 1, 2.5], {K, 70, 130},
  PlotLabel -> "Smile for Elasticity + Spread Model of Put\n"];
```

Smile for Elasticity + Spread Model of Put



These are qualitatively similar to what one sees in equity markets (note that we regard the FX case as totally different - it is not a candidate for this type of elasticity treatment). We see that the effect of the elasticity in combination with the spread is to dramatically raise the OTM implied volatilities, and to flatten or curve the ITM region. It is clear that the SRCEV model does capture part of the skew effect seen in equity markets. The reader should note the discussion by R.J. Brenner, "Volatility is not constant", for further comments on CEV and related level-dependent volatility models in the book by Nelken (1990). The time-dependence of the elasticity and spread-adjusted implied volatility is that the smile becomes more of a skew, and the overall level drifts down - here is the implied volatility surface for 0.5 to 5 years:

```
Plot3D[PutForwardBackward[100, K, 0.05, 0, 0.2, t, 2.5],
  {K, 70, 130}, {t, 0.5, 5},
  PlotLabel -> "Vol Surface for Elasticity\n+ Spread Model of Put\n"];
```



If, in addition, one adds in the effect of a time-dependent yield curve, but uses a constant interest rate in assessing BS implied volatility, one can generate a fictitious term structure for the implied volatility. The lesson that should be learned is that one should subtract out all the known time-dependent and asset-level-dependent effects before attempting to build two-factor stochastic or GARCH models of volatility.

28.11 Closing Remarks

A book has to end somewhere, and this is where matters are brought to a close. There is much more that could be said about CEV and its cousins in interest-rate modelling. An important point is that the sequence of transformations made to analyse the PDE can also be applied, with minor modifications, to the family of interest-rate models discussed in Chapters 25-27. All such single-factor models, and the CEV family, can be reduced to the standard diffusion equation with advection and source terms, allowing most of the numerical machinery developed in Chapters 13-20 to be brought to bear, in a single unified framework. Another point not discussed here is how to characterize the probability distributions that arise when ν is not unity and there is also mean reversion. It is left as an exercise for the reader to follow through the sequence of steps of Section 28.3 with the mean reversion included. The resulting PDE becomes nice (in a sense that becomes obvious when you try it) only if $\nu = 1/2$, 1 or the mean-reversion is excluded. What the enveloping distribution is when mean-reversion is present and ν is general is not known to this author.

Chapter 28 Bibliography

Beckers, S., 1980, The constant elasticity of variance model and its implications for option pricing *Journal of Finance*, 35, June, p. 661.

Cox, J., 1996, The constant elasticity of variance option pricing model, *Journal of Portfolio Management*, special issue.

Hauser, S., and Lauterbach, B., 1996, Tests of warrant pricing models: the trading profits perspective, *Journal of Derivatives*, Winter, p. 71.

Jarrow, R.A. and Rudd, A., 1983, *Option Pricing* (Section 11.6), Irwin.

Lauterbach, B. and Schultz, P., 1990, Pricing warrants: an empirical study of the Black-Scholes model and its alternatives, *Journal of Finance,* 45 (Sept.), p. 1181;

Nelken, I. (editor), 1990, *The Handbook of Exotic Options*, Irwin.

Press, W.H., Teukolsky, S.A., Vetterling, W.T. and Flannery, B.P., 1992, *Numerical Recipes in C - the Art of Scientific Computing*, 2nd edition. Cambridge University Press.

Schroder, M., 1989, Computing the constant elasticity of variance option pricing formula, *Journal of Finance,* 44, March, p. 221.